Demo # 1 completo

Crear escenario
1. Cree un proyecto vacío

2. Menú principal 

    Project

        Resources…


Seleccionar “FishArt”


    Presionar “Add to Project”


         Presionar “Save”

3. Panel derecho

   Create

       Static Sprites


Arrastre “BackGroundImage” y acomode
   Edit

       Scene Object

Layer = 30

   Create

       Static Sprites


Arrastre “RocksFarImage” y acomode

   Edit

       Scene Object


Layer = 20

   Create

       Static Sprites


Arrastre “RocksNearImage” y acomode

   Edit

       Scene Object


Layer = 10

   Create

       Animated Sprites


Arrastre “WaveImage” y acomode (rodee todo el escenario)
   Edit

       Scene Object


Layer = 5

       Scroller



Scroll Speed



    X = 4.000


       Blending


Enabled = check



Alpha  =  100 (modifique a su gusto la transparencia)

Crear Jugador
1.    Create

       Animated Sprites

      
Arrastre “Angelfish1SheetImageMapAnimation”
         Edit

       Scene Object


Position


   X = 0 & Y = 0


Size


   Width = 20 & Height = 18

Layer = 11


   Scripting


       Class = PlayerFish


   Collision


       Send Collision = check


       Recive Collision = check


       Detection mode = POLYGON


       Superscribe ellipse = check


   World limits


        Limit mode = CLAMP


   Dynamic Fields


        Llene los campos y presione la cruz verde para terminar cada renglón



vSpeed = 15



hSpeed = 30



maxHeight = 20



life = 100



maxWidth = 22





lifeDrain = -2

2. Seleccionada la imagen pase el cursor sobre la parte superior de la misma y elija la herramienta


   Edit Collision Polygon


       Punto 0



X = 0.854   &   Y = 0.176


       Punto 1



X = 0.982   &   Y = 0.206

       Punto 2



X = 0.982   &   Y = 0.342

       Punto 3



X = 0.854   &   Y = 0.372


Save


   Change World limits


       Rodee el escenario desde la orilla de arriba y ajustando a los lados con el borde del escenario, luego deje un espacio antes de llegar al borde de abajo del escenario para que el pez no pase por debajo de las piedras, esto ayuda al efecto visual

Crear los enemigos

1.    Create

       Animated Sprites

      
Arrastre “Angelfish2SheetImageMapAnimation”

         Edit

       Scene Object


Position


   Esquina superior derecha


Size


   Width = 20 & Height = 18


Layer = 16


   Scripting


       Class = Fish


   Collision


       Recive Collision = Check


       Detection mode = FULL

   World limits


        Limit mode = NULL

        Callback =  Check


   Dynamic Fields


        Llene los campos y presione la cruz verde para terminar cada renglón



maxSpeed = 30


minSpeed = 5

2. Seleccionada la imagen pase el cursor sobre la parte superior de la misma y elija la herramienta


   Edit Collision Polygon


       Punto 0



X = 0.859   &   Y = 0.201


       Punto 1



X = 0.994   &   Y = 0.221


       Punto 2



X = 1.000   &   Y = 0.334


       Punto 3



X = 0.834   &   Y = 0.358



Save

3. Para crear mas enemigos solo haga lo siguiente cuantos enemigos quiera que existan

Menú principal 


Edit


   Copy  & Paste



Después de pegar mueva el enemigo a la posición en la que quedará, puede si lo prefiere cambiar la apariencia del enemigo y recuerde corregir el borde de colisión de la boca ya que cada apariencia de pez tiene la boca en una distinta posición

Para cambiar la apariencia

     Animated Sprite


Animation


    Del menú desplegable seleccione la apariencia que prefiera
Crear un alimento para los peces
1. Antes de comenzar haga click derecho sobre la imagen numerada como 1/14 “FishImagesImage”
2.   Create

       Static Sprites


Arrastre “FishImagesImage = burbuja” y acomode

   Edit

      Static Sprites     
           Frame = 13
      Scene Object

Position


   Centro en la parte superior fuera del escenario
Size
              Width = 3 & Height = 3


Layer = 12

   Scripting


       Class = FishFood

   Collision


       Send Collision = check


       Callback = check


       Detection mode = CIRCLE

   World limits


        Limit mode = NULL

        Callback = Check


   Dynamic Fields


        Llene los campos y presione la cruz verde para terminar cada renglón



lifeValue = 10


maxSpeed = 15


minSpeed = 5
3. cambie los limites del mundo de la burbuja como la imagen

[image: image2.png]



Creando Minas
1. Antes de comenzar haga doble click derecho sobre la imagen numerada como 1/14 “FishImagesImage”
2.   Create

       Static Sprites


Arrastre “FishImagesImage = Mina” y acomode

   Edit

      Static Sprites     
           Frame = 12

      Scene Object

Position


   Centro en la parte superior fuera del escenario (a un lado de la burbuja)
Size
              Width = 4 & Height = 4


Layer = 12

   Scripting


       Class = Mine

   Collision


       Send Collision = check


       Callback = check


       Detection mode = CIRCLE

   Dynamic Fields


        Llene los campos y presione la cruz verde para terminar cada renglón



minX = -40


maxX =  40


minY = 5


maxY = 20



damageValue = 30



minRespawnTime = 10000



maxRespawnTime = 20000



minFloatRange = 5



maxFloatRange = 15



minSpeed = 4


maxSpeed = 7
Scripts
Cree o reemplace los siguientes scripts

Archivo “game.cs”

//---------------------------------------------------------------------------------------------

// Torque Game Builder

// Copyright (C) GarageGames.com, Inc.

//---------------------------------------------------------------------------------------------

//---------------------------------------------------------------------------------------------

// startGame

// All game logic should be set up here. This will be called by the level builder when you

// select "Run Game" or by the startup process of your game to load the first level.

//---------------------------------------------------------------------------------------------

function startGame(%level)

{

   exec("./player.cs");

   exec("./fishfood.cs");

   exec("./fish.cs");    

   exec("./mine.cs");

   Canvas.setContent(mainScreenGui);

   Canvas.setCursor(DefaultCursor);

   new ActionMap(moveMap);   

   moveMap.push();

   $enableDirectInput = true;

   activateDirectInput();

   enableJoystick();

   sceneWindow2D.loadLevel(%level);

}

//---------------------------------------------------------------------------------------------

// endGame

// Game cleanup should be done here.

//---------------------------------------------------------------------------------------------

function endGame()

{

   sceneWindow2D.endLevel();

   moveMap.pop();

   moveMap.delete();

}

Archivo “Player.cs”

function PlayerFish::onLevelLoaded(%this, %scenegraph)

{

   $FishPlayer = %this;

   moveMap.bindCmd(keyboard, "Up", "fishPlayerUp();", "fishPlayerUpStop();");

   moveMap.bindCmd(keyboard, "Down", "fishPlayerDown();", "fishPlayerDownStop();");

   moveMap.bindCmd(keyboard, "Left", "fishPlayerLeft();", "fishPlayerLeftStop();");

   moveMap.bindCmd(keyboard, "Right", "fishPlayerRight();", "fishPlayerRightStop();");

   moveMap.bindCmd(keyboard, "space", "fishPlayerBoost();", "fishPlayerBoostStop();");

   %this.lowerLife();

}

function PlayerFish::updateMovement(%this)

{

   if(%this.dead)

      return;

   if(%this.moveLeft)

   {

      $FishPlayer.setFlipX(true);

      $FishPlayer.setLinearVelocityX( -$FishPlayer.hSpeed );

   }

   if(%this.moveRight)

   {

      $FishPlayer.setFlipX(false);

      $FishPlayer.setLinearVelocityX( $FishPlayer.hSpeed );

   }

   if(%this.moveUp)

   {

      %this.setLinearVelocityY( -$FishPlayer.vSpeed );

   }

   if(%this.moveDown)

   {

      %this.setLinearVelocityY( $FishPlayer.vSpeed );

   }

   if(!%this.moveLeft && !%this.moveRight)

   {

      %this.setLinearVelocityX( 0 );

   }

   if(!%this.moveUp && !%this.moveDown)

   {

      %this.setLinearVelocityY( 0 );

   }

}

function PlayerFish::modifyLife(%this, %dmg)

{

   %this.life += %dmg;

   if(%this.life > 100)

   {

      %this.life = 100;

   } else if (%this.life < 0)

   {

      %this.life = 0;

   }

   if(%this.life <= 30)

   {

      %this.dead();

   } else

   {

      %this.updateLifeSize();

   }

}

function PlayerFish::updateLifeSize(%this)

{

   %lifeMultiplier = %this.life / 100;

   %newWidth = %this.maxWidth * %lifeMultiplier;

   %newHeight = %this.maxHeight * %lifeMultiplier;

   %this.setSize(%newWidth, %newHeight);

}

function PlayerFish::dead(%this)

{

   %this.setFlipY(true);

   %this.setLinearVelocityY(-10);

   %this.dead = true;

}

function PlayerFish::lowerLife(%this)

{

   %this.modifyLife(%this.lifeDrain);

   if(!%this.dead)

   {

      %this.schedule(500, "lowerLife");

   }

}

function fishPlayerUp()

{

   $FishPlayer.moveUp = true;

   $FishPlayer.updateMovement();

}

function fishPlayerDown()

{

   $FishPlayer.moveDown = true;

   $FishPlayer.updateMovement();

}

function fishPlayerLeft()

{

   $FishPlayer.moveLeft = true;

   $FishPlayer.updateMovement();

}

function fishPlayerRight()

{

   $FishPlayer.moveRight = true;

   $FishPlayer.updateMovement();

}

function fishPlayerUpStop()

{

   $FishPlayer.moveUp = false;

   $FishPlayer.updateMovement();

}

function fishPlayerDownStop()

{

   $FishPlayer.moveDown = false;

   $FishPlayer.updateMovement();

}

function fishPlayerLeftStop()

{

   $FishPlayer.moveLeft = false;

   $FishPlayer.updateMovement();

}

function fishPlayerRightStop()

{

   $FishPlayer.moveRight = false;

   $FishPlayer.updateMovement();

}

function fishPlayerBoost()

{

   if($FishPlayer.dead)

      return;   

   %flipX = $FishPlayer.getFlipX();

   if(!%flipX)

   {

      %hSpeed = $FishPlayer.hSpeed * 3;

   } else

   {

      %hSpeed = -$FishPlayer.hSpeed * 3;

   }

   $FishPlayer.setLinearVelocityX(%hSpeed);

}

function fishPlayerBoostStop()

{

   $FishPlayer.setLinearVelocityX(0);

}

Archivo “FishFood.cs”
function FishFood::onLevelLoaded(%this, %scenegraph)

{

   %this.startPositionY = %this.getPositionY();

   %this.setLinearVelocityY(getRandom(%this.minSpeed, %this.maxSpeed));

}

function FishFood::onWorldLimit(%this, %mode, %limit)

{

   if(%limit $= "bottom")

   {

      %this.spawn();

   }

}

function FishFood::spawn(%this)

{

   %this.setPosition(getRandom(-40, 40), %this.startPositionY);

   %this.setLinearVelocityY(getRandom(%this.minSpeed, %this.maxSpeed));

}

function FishFood::onCollision(%srcObj, %dstObj, %srcRef, %dstRef, %time, %normal, %contactCount, %contacts)

{

   if(%dstObj.class $= "PlayerFish")

   {

      %srcObj.spawn();

      %dstObj.modifyLife(%srcObj.lifeValue);

   } else if (%dstObj.class $= "Fish")

   {

      %srcObj.spawn();

   }

}
Archivo “Fish.cs”

function Fish::getSpeed(%this)

{

   return getRandom(%this.minSpeed, %this.maxSpeed);

}

function Fish::reposition(%this)

{

   %this.setPositionY(getRandom(-30, 20));

}

function Fish::onLevelLoaded(%this, %scenegraph)

{

   %this.setLinearVelocityX(%this.getSpeed());

}

function Fish::onWorldLimit(%this, %mode, %limit)

{

   switch$(%limit)

   {

      case "left":

         %this.setFlipX(false);

         %this.setLinearVelocityX(%this.getSpeed());

         %this.reposition();

      case "right":

         %this.setFlipX(true);

         %this.setLinearVelocityX(-%this.getSpeed());

         %this.reposition();

   }

}
Archivo “Mine.cs”

function Mine::onLevelLoaded(%this, %scenegraph)

{

   %this.startingY = %this.getPositionY();

   %this.spawn();

}

function Mine::spawn(%this)

{

   %this.state = "fallingDown";

   %this.speed = getRandom(%this.minSpeed, %this.maxSpeed);

   %this.originX = getRandom(%this.minX, %this.maxX);

   %this.originY = getRandom(%this.minY, %this.maxY);

   %this.setPosition(%this.originX, %this.startingY);

   %this.moveTo(%this.originX, %this.originY, %this.speed, true, true);

if(isEventPending(%this.respawnSchedule))

      cancel(%this.respawnSchedule);

   %respawnTime = getRandom(%this.minRespawnTime, %this.maxRespawnTime);

   %this.respawnschedule = %this.schedule(%respawnTime, "spawn");

}

function Mine::onPositionTarget(%this)

{

   %this.floatRange = getRandom(%this.minFloatRange, %this.maxFloatRange);

   switch$(%this.state)

   {

      case "fallingDown":

         %this.floatDown();

      case "floatingDown":

         %this.floatUp();

      case "floatingUp":

         %this.floatDown();

   }

}

function Mine::floatUp(%this)

{

   %this.state = "floatingUp";

   %targetY = %this.originY - %this.floatRange;

   %this.moveTo(%this.originX, %targetY, %this.speed, true, true);

}

function Mine::floatDown(%this)

{

   %this.state = "floatingDown";

   %targetY = %this.originY + %this.floatRange;

   %this.moveTo(%this.originX, %targetY, %this.speed, true, true);

}

function Mine::onCollision(%srcObj, %dstObj, %srcRef, %dstRef, %time, %normal, %contactCount, %contacts)

{

   if(%dstObj.class $= "PlayerFish")

   {

      %srcObj.spawn();

      %dstObj.modifyLife(-%srcObj.damageValue);

   }

}

